```
 ADDR  CODE      STMT SOURCE STATEMENT

                 0001          NAME    DBSLDR
                 0003 ;*********************************************************
                 0004 ;                                                        *
                 0005 ;        PROGRAM ID:      DDBIOS LOADER                  *
                 0006 ;                                                        *
                 0007 ;        VERSION:         2.2          RELEASE 2         *
                 0008 ;                                                        *
                 0009 ;*********************************************************
                 0010 ;                                                        *
                 0011 ;        PROPERTY OF:     JADE COMPUTER PRODUCTS         *
                 0012 ;                         4901 W. ROSECRANS BLVD.        *
                 0013 ;                         HAWTHORNE, CALIFORNIA          *
                 0014 ;                         90250,   U.S.A.                *
                 0015 ;                                                        *
                 0016 ;*********************************************************
                 0017 ; THE BIOS LOADER IS READ INTO THE DCM SECTOR BUFFER      *
                 0018 ; AFTER DCM HAS INITIALIZED. THE BIOS LOADER PROGRAM      *
                 0019 ; IS THEN  EXECUTED  WHICH READS THE  DDBIOS  MODULE      *
                 0020 ; INTO BANK 1.  THE COMMAND BLOCK (IN DCM) IS SET TO      *
                 0021 ; INDICATE  DDBIOS  MODULE SIZE AND THE  SYSTEM LOAD      *
                 0022 ; ADDRESS.  THE  BIOS LOADER PROGRAM IS GENERATED BY      *
                 0023 ; MOVCPM.COM AS THE COLD START LOADER (900-97F HEX).      *
                 0024 ; THIS MODULE IS PROVIDED FOR REFERENCE PURPOSES.         *
                 0025 ;*********************************************************
                 0026 ; THE  DDBIOS LOADER IS COMPATABLE WITH  REV B AND C      *
                 0027 ; DOUBLE D CONTROLLER BOARDS.  IT IS COMPATABLE WITH      *
                 0028 ; FD1791-01 / FD1793-01.  IT WILL ALSO FUNCTION WITH      *
                 0029 ; THE CURRENT FD179X-02 SERIES.                           *
                 0030 ;*********************************************************
```

```
 ADDR  CODE      STMT SOURCE STATEMENT

                 0032 ;*********************************************************
                 0033 ; CONTROLLER PORT ASSIGNMENTS                             *
                 0034 ;*********************************************************
                 0035
>0000            0036 BL$STS   EQU     000H    ;BOARD STATUS
>0000            0037 BL$CTL   EQU     000H    ;BOARD CONTROLS
>0004            0038 WD$CMD   EQU     004H    ;179X-02 COMMAND REGISTER
>0004            0039 WD$STS   EQU     004H    ;179X-02 STATUS REGISTER
>0006            0040 WD$SEC   EQU     006H    ;179X-02 SECTOR REGISTOR
>0007            0041 WD$DTA   EQU     007H    ;179X-02 DATA REGISTER
>0010            0042 XP$MTO   EQU     010H    ;MOTOR TIME OUT
>0040            0043 XP$MTX   EQU     040H    ;MOTOR TIME EXTEND
>0080            0044 XP$DSH   EQU     080H    ;DATA SYNC HOLD
                 0045
                 0046 ;*********************************************************
                 0047 ; 179X-02 COMMAND AND MASK.                               *
                 0048 ;*********************************************************
                 0049
>0088            0050 DC$RDS   EQU   10001000B  ;READ SECTOR.
>009C            0051 DM$RER   EQU   10011100B  ;READ ERROR MASK.
                 0052
                 0053 ;*********************************************************
                 0054 ; SYSTEM ASSIGNMENTS                                      *
                 0055 ;*********************************************************
                 0056
>0014            0057 NMBR$K   EQU     20              ;SYSTEM SIZE IN K.
>0400            0058 LNG$1K   EQU     1024            ;TOTAL BYTES IN 1K.
>5000            0059 CPM$SZ   EQU     NMBR$K*LNG$1K   ;TOTAL SYSTEM BYTES.
>0600            0060 BIOS$S   EQU     LNG$1K*3/2      ;BIOS ALLOCATED SIZE.
>4A00            0061 BIOS$A   EQU     CPM$SZ-BIOS$S   ;BIOS LOAD ADDRESS.
                 0062
                 0063 ;*********************************************************
                 0064 ; INTERNAL MEMORY ASSIGNMENTS                             *
                 0065 ;*********************************************************
                 0066
>1000            0067 BANK$0   EQU     1000H           ;LOWER BANK ADDRESS.
>0400            0068 BANK$L   EQU     0400H           ;1K BANK LENGTH.
>1400            0069 BANK$1   EQU     BANK$0+BANK$L   ;UPPER BANK ADDRESS.
>1370            0070 IO$BLK   EQU     BANK$0+0370H    ;I/O BLOCK ADDRESS.
>1377            0071 CB$STS   EQU     IO$BLK+0007H    ;COMMAND STATUS BYTE.
>1378            0072 CW$LAD   EQU     IO$BLK+0008H    ;BIOS LOAD ADDR LOC.
>137A            0073 CW$LNG   EQU     IO$BLK+000AH    ;BIOS LOAD LENGTH LOC.
>1380            0074 SEC$BF   EQU     BANK$0+0380H    ;SECTOR BUFFER AREA.
                 0075
                 0076 ;*********************************************************
                 0077 ; BIOS PROGRAM LINKAGE.                                   *
                 0078 ;*********************************************************
                 0079
>0004            0080 SEC$BG   EQU     4               ;FIRST BIOS SECTOR.
>0008            0081 SEC$NM   EQU     8               ;NUMBER OF SECTORS.
>000B            0082 SEC$EX   EQU     SEC$BG+SEC$NM-1 ;LAST BIOS SECTOR.
                 0083
                 0084 ;*********************************************************
```

```
 ADDR  CODE      STMT SOURCE STATEMENT

                 0086 ;*****************************************************
                 0087 ; ASSEMBLER DIRECTIVES                                *
                 0088 ;*****************************************************
                 0089
                 0090         PSECT   ABS             ;ABSOLUTE ADDRESSING.
 >1380           0091         ORG     SEC$BF          ;PROGRAM START POINT.
                 0092
                 0093 ;*****************************************************
                 0094 ; INITIALIZE BIOS READ OPERATION                      *
                 0095 ;*****************************************************
                 0096
 1380  210004    0097 BEGIN:  LD      HL,LNG$1K       ;BIOS LOAD LENGTH.
 1383  227A13    0098         LD      (CW$LNG),HL     ;LOAD LENGTH SET.
 1386  21004A    0099         LD      HL,BIOS$A       ;BIOS SYSTEM ADDR.
 1389  227813    0100         LD      (CW$LAD),HL     ;LOAD ADDRESS SET.
 138C  210014    0101         LD      HL,BANK$1       ;BIOS LOAD POINT.
                 0102
                 0103 ;*****************************************************
                 0104 ; SET-UP FOR EACH READ SECTOR COMMAND                  *
                 0105 ;*****************************************************
                 0106
 138F  FD21A813  0107 RD$SEC: LD      IY,RD$TST       ;SET NMI VECTOR.
 1393  3AC413    0108         LD      A,(SECTOR)      ;FIRST BIOS SECTOR.
 1396  A9        0109         XOR     C               ;INVERT (1791-01).
 1397  D306      0110         OUT     (WD$SEC),A      ;SET 179X-02 SEC REG.
 1399  3E88      0111         LD      A,DC$RDS        ;READ SECTOR CMND.
 139B  A9        0112         XOR     C               ;INVERT (1791-01).
 139C  D304      0113         OUT     (WD$CMD),A      ;ISSUE 179X-02 COMMAND.
                 0114
                 0115 ;*****************************************************
                 0116 ; READ SECTOR OPERATION                               *
                 0117 ;*****************************************************
                 0118
 139E  DB80      0119 RD$BYT: IN      A,(XP$DSH)      ;WAIT FOR DATA.
 13A0  DB07      0120         IN      A,(WD$DTA)      ;INPUT INV DATA.
 13A2  A9        0121         XOR     C               ;INVERT (1791-01).
 13A3  77        0122         LD      (HL),A          ;STORE DCM BYTE.
 13A4  23        0123         INC     HL              ;INCREMENT POINTER.
 13A5  C39E13    0124         JP      RD$BYT          ;REPEAT OPERATION.
                 0125
                 0126 ;*****************************************************
```

```
 ADDR  CODE      STMT SOURCE STATEMENT

               0128 ;*********************************************************
               0129 ; CHECK READ SECTOR STATUS, REPEAT UNTIL BIOS LOADED *
               0130 ;*********************************************************
               0131
 13A8  E69C    0132 RD$TST: AND     DM$RER          ;TEST FOR ERRORS.
 13AA  200D    0133         JR      NZ,ERRORS       ;ERROR DETECTED.
 13AC  3AC413  0134         LD      A,(SECTOR)      ;GET SECTOR NMBR.
 13AF  FE0B    0135         CP      SEC$EX          ;CHECK IF LAST SEC.
 13B1  280F    0136         JR      Z,FINISH        ;GO IF FINISHED.
 13B3  3C      0137         INC     A               ;INCREMENT.
 13B4  32C413  0138         LD      (SECTOR),A      ;STORE SECTOR NUMBER.
 13B7  18D6    0139         JR      RD$SEC          ;READ NEXT SECTOR.
               0140
               0141 ;*********************************************************
               0142 ; READ ERROR HAS BEEN DETECTED                          *
               0143 ;*********************************************************
               0144
 13B9  327713  0145 ERRORS: LD      (CB$STS),A      ;DISPLAY ERROR STATUS.
 13BC  AF      0146         XOR     A               ;ZERO  A REGISTER.
 13BD  D300    0147         OUT     (BL$CTL),A      ;DESELECT DRIVE.
 13BF  DB10    0148         IN      A,(XP$MTO)      ;MOTOR OFF!
 13C1  76      0149         HALT                    ;TERMINATE.
               0150
               0151 ;*********************************************************
               0152 ; BIOS SECTOR HAVE BEEN LOADED                           *
               0153 ;*********************************************************
               0154
 13C2  FB      0155 FINISH: EI                      ;ENABLE INTERRUPTS.
 13C3  76      0156         HALT                    ;SHUTDOWN BOARD.
               0157
               0158 ;*********************************************************
               0159 ; SECTOR NUMBER STORAGE                                  *
               0160 ;*********************************************************
               0161
 13C4  04      0162 SECTOR: DEFB    SEC$BG          ;SECTOR COUNTER.
               0163
               0164 ;*********************************************************
               0165         END
```

ADDR CODE STMT SOURCE STATEMENT

CROSS REFERENCE LISTING

SYMBOL	VALUE	TYPE	STMT	STATEMENT REFERENCES
BANK$0	1000		0067	0074 0070 0069
BANK$1	1400		0069	0101
BANK$L	0400		0068	0069
BEGIN	1380		0097	
BIOS$A	4A00		0061	0099
BIOS$S	0600		0060	0061
BL$CTL	0000		0037	0147
BL$STS	0000		0036	
CB$STS	1377		0071	0145
CPM$SZ	5000		0059	0061
CW$LAD	1378		0072	0100
CW$LNG	137A		0073	0098
DC$RDS	0088		0050	0111
DM$RER	009C		0051	0132
ERRORS	13B9		0145	0133
FINISH	13C2		0155	0136
IO$BLK	1370		0070	0073 0072 0071
LNG$1K	0400		0058	0097 0060 0059
NMBR$K	0014		0057	0059
RD$BYT	139E		0119	0124
RD$SEC	138F		0107	0139
RD$TST	13A8		0132	0107
SEC$BF	1380		0074	0091
SEC$BG	0004		0080	0162 0082
SEC$EX	000B		0082	0135
SEC$NM	0008		0081	0082
SECTOR	13C4		0162	0138 0134 0108
WD$CMD	0004		0038	0113
WD$DTA	0007		0041	0120
WD$SEC	0006		0040	0110
WD$STS	0004		0039	
XP$DSH	0080		0044	0119
XP$MTO	0010		0042	0148
XP$MTX	0040		0043	

ERRORS=0000